
AB_Bias_Audio_OffsetNoise
*   _________________________________________________________________________________
*  |             |       |     |           |         VCC           |           |     |
*  |            <V>VNO5  |    <V>VNO6      |                      <V>VNO7      |     |
*  |        QP10 |       |     |  VCC      |                       |           |    _|_
*   ->         <-      <-      |           |                       |           |   /VCC\
*  QP9 `|____|'______|' QP11    ->       <-                         ->       <-    \___/
*    _ '| |  |`__    |`__      QP3`|___|'QP44                      QP6`|___|'QP7    _|_
*   |     |      |       |      _ '| | |`_                          _ '| | |`_      ///
*   |_____|      |       |     |     |____|VTA2                    |     |____|VTA4
*  _|_ VBIAS     |       |     |          |                        |          |
* / _ \          |       |     |         <-                        |         <-
* \/ \/          |       |     |_______|' QP5        ______________|_______|' QP8
* /\_/\ IBIAS    |       |     |VTA1   |`_          |               VTA3   |`_
* \___/          |       |     |          |         |                         | OUT
*  _|_         __|       |__   |          |________/|\____________            |
*  \\\        |             |  |                    |             |           |
*             |             |  |           _________|             |           |
*             |  VBN1       |  |          |             VBN6      |           | OUT
*             |___________  |_/|\________/|\________________      |           |
*    ___      | QN1 |  QN2|    |          |     |QN5   |QN6 |     |           |___/\  /\  /\_
*  _|_  |     |    _|    _|   _|          |_    |_     |_   |     |           |     \/  \/   |
*  ///  |     |__|'____|'___|'   QN3   QN4  `|____`|_____`|_|     |           |             _|_
*      _|_    MX |`->  |`-> |`->          <-'|  <-'|   <-'|       |           |             ///
*     /VEE\         |     |    |NX      NX|     |1X    |MX        | VTA5     _|
*     \___/        <V>   <V>  <V>         |     |      |          |________|'QN9
*      _|_     VNO2 | VNO3|VNO4|          |_____|      |          |        |`->
*      \ /          |     |____|______     VEN4 |      |          |     VT6   |
*       V           |      VEN3       |         |      |          |     ______|
*                   |________________/|\________|      |          |_    |    _|
*                   |                 |                |         QN7`|__|__|'QN8
*                  <V>VNO1            |                |          <-'|     |`->
*                   |                 |________________|          |           |
*                 <-                                   ->        <V>VNO8      |
*            ___|'XQP1                                QP2 `|_     |           |
*          _|_  |`_                                     _ '| |   _|_         _|_
*         /VIN\     |                                  |    _|_  \ /         \ /
*         \___/     |                                 _|    ///   V  VEE      V
*           |      _|_                                \ /
*          _|_     \ /  VEE                            V VEE
*          ///      V
*

*
.OPTIONS  GMIN=1e-12    METHOD=trap   srcsteps = 1  gminsteps = 1
.OPTIONS  ITL1=400      ITL2=100      ITL6=100 
.OPTIONS  RELTOL=.001   ABSTOL=1p     VNTOL=1u

*======== ====== ====== ====== ====== ====== ====== ====== ======
VCC       VCC    0      DC     2
VEE       0      VEE    DC     2
VIN       VIN    0      DC     0  
IBIAS     VBIAS  0      .16u
QP9       VBIAS  VBIAS  VCC    PNP1   1
QP10      VBN1   VBIAS  VCCB   PNP1   1
QP11      VBN6   VBIAS  VCC    PNP1   1
QP1       VEE    VIN    VEN4B  PNP1   1
QP2       VEE    0      VEN3   PNP1   1
QP3       VTA1   VTA2   VCCC   PNP1   1
QP4       VTA2   VTA2   VCC    PNP1   1
QP5       VTA5   VTA1   VTA2   PNP1   1
QP6       VTA3   VTA7   VCCD   PNP1   1
QP7       VTA7   VTA7   VCC    PNP1   1
QP8       OUT    VTA3   VTA7   PNP1   1

QN7       VTA5   VTA6   VEEB   NPN1   1
QN8       VTA6   VTA6   VEE    NPN1   1
QN9       OUT    VTA5   VTA6   NPN1   1

R1        OUT    0      1
QN1       VBN1   VBN1   VEN4C  NPN1   4.5
QN6       VBN6   VBN6   VEN3   NPN1   4.5
QN2       VBN1   VBN1   VEN3B  NPN1   1
QN5       VBN6   VBN6   VEN4   NPN1   1
QN3       VTA1   VBN1   VEN3C  NPN1   15
QN4       VTA3   VBN6   VEN4   NPN1   15

VNO1      VEN4   VEN4B  DC     0
VNO2      VEN4   VEN4C  DC     0
VNO3      VEN3   VEN3B  DC     0
VNO4      VEN3   VEN3C  DC     0
VNO5      VCC    VCCB   DC     0
VNO6      VCC    VCCC   DC     0



VNO7      VCC    VCCD   DC     0
VNO8      VEE    VEEB   DC     0

.control
setplot new
set thisData =       data
let $thisData =      0 * vector(38)
set thisData2 =      data2
let $thisData2 =     0 * vector(38)

let n=0
repeat 17
let        unknown.{$thisData2}[n] = n
let n=n+1
endrepeat

*==========Geometry_Sensitivity===================
* Add a voltage source in series with one transistor for each matched pair
* measure the effect of adding a standart error (1mV or 1%)
* Store all gain and offset results for each transistor
* Calibrate each transistor to its area and processing expectations

*#0========Define_1X_NPN_and_PNPL_and_RB_Offsett===========
let        npnoff = .6m
let        pnpoff = 1.1m
let        rboff  = .005
let        unknown.{$thisData}[0] = npnoff
let        unknown.{$thisData}[1] = pnpoff
let        unknown.{$thisData}[2] = rboff
let        npnbf = @NPN1[bf]
let        pnpbf = @PNP1[bf]
let        npnrb = @NPN1[rb]
let        pnprb = @PNP1[rb]

echo      "Process Expectations  "
echo      "NPN_offset   =  $&npnoff  "
echo      "NPN_beta     =  $&npnbf  "
echo      "NPN_rbase    =  $&npnrb  "
echo      "PNPL_offset  =  $&pnpoff  "
echo      "PNPL_beta    =  $&pnpbf  "
echo      "PNPL_rbase   =  $&pnprb  "
echo      "Rbase_offset =  $&rboff  "
echo      ""

*#0========Measure_With_No_Errors===========
op
let        unknown.{$thisData}[3] = out
let        ibb    = @ibias[dc]
let        icc    =  -1*vcc#branch
echo      "Ibias        =  $&ibb  "
echo      "Isupply      =  $&icc "
echo      ""

*==========Measure_Offset_Gain_Each_Transitor===================
* Add 1mv offset one at a time to each transistor
* Measure change at output
* Store gain and offset results for each transistor
* measure transistor area
* calculate each transistors offset from area and npnoff, pnpoff, and rboff

*#1========Measure_OTA_gm_At_1mV===========
alter      vin    dc =    1m
op
let        gm =   abs(out- unknown.{$thisData}[3]) /1m
let        unknown.{$thisData}[4] = gm
alter      vin    dc =    0

*#2========Measure_VNO1_effect===========
alter      vno1   dc =    1m
op
let        procOff = unknown.{$thisData}[1]
let        Tarea   = @qp1[area]
let        Ioutref = unknown.{$thisData}[3]
let        gm      = unknown.{$thisData}[4]
let        vingain  = 1000*(out - Ioutref)/gm
let        pairOff = (1/Tarea)^.5*procOff
let        vinoff  = pairOff*vingain
let        unknown.{$thisData}[5] = vingain  



let        unknown.{$thisData}[6] = vinoff
echo      "Pair      ProcOff   Area  Pair_Off  Gain2Vin   Vin_Offset "
echo      "QP1/QP2   $&procOff    $&Tarea     $&pairOff    $&vingain    $&vinoff"

*#3========Measure_VNO2_effect===========
alter      vno1   dc =    0
alter      vno2   dc =    1m
op
let        procOff = unknown.{$thisData}[0]
let        Tarea   = @qn1[area]
let        Ioutref = unknown.{$thisData}[3]
let        gm      = unknown.{$thisData}[4]
let        vingain  = 1000*(out - Ioutref)/gm
let        pairOff = (1/Tarea)^.5*procOff
let        vinoff  = pairOff*vingain
let        unknown.{$thisData}[7] = vingain  
let        unknown.{$thisData}[8] = vinoff
echo      "QN1/QN6   $&procOff    $&Tarea     $&pairOff    $&vingain   $&vinoff"

*#4========Measure_VNO3_effect===========
alter      vno2   dc =    0
alter      vno3   dc =    1m
op
let        procOff = unknown.{$thisData}[0]
let        Tarea   = @qn2[area]
let        Ioutref = unknown.{$thisData}[3]
let        gm      = unknown.{$thisData}[4]
let        vingain  = 1000*(out - Ioutref)/gm
let        pairOff = (1/Tarea)^.5*procOff
let        vinoff  = pairOff*vingain
let        unknown.{$thisData}[9] = vingain  
let        unknown.{$thisData}[10] = vinoff
echo      "QN2/QN5   $&procOff    $&Tarea     $&pairOff   $&vingain  $&vinoff"

*#5========Measure_VNO4_effect===========
alter      vno3   dc =    0
alter      vno4   dc =    1m
op
let        procOff = unknown.{$thisData}[0]
let        Tarea   = @qn3[area]
let        Ioutref = unknown.{$thisData}[3]
let        gm      = unknown.{$thisData}[4]
let        vingain  = 1000*(out - Ioutref)/gm
let        pairOff = (1/Tarea)^.5*procOff
let        vinoff  = pairOff*vingain
let        unknown.{$thisData}[11] = vingain  
let        unknown.{$thisData}[12] = vinoff
echo      "QN3/QN4   $&procOff    $&Tarea     $&pairOff   $&vingain   $&vinoff"

*#6========Measure_VNO5_effect===========
alter      vno4   dc =    0
alter      vno5   dc =    1m
op
let        procOff = unknown.{$thisData}[1]
let        Tarea   = @qp10[area]
let        Ioutref = unknown.{$thisData}[3]
let        gm      = unknown.{$thisData}[4]
let        vingain  = 1000*(out - Ioutref)/gm
let        pairOff = (1/Tarea)^.5*procOff
let        vinoff  = pairOff*vingain
let        unknown.{$thisData}[13] = vingain  
let        unknown.{$thisData}[14] = vinoff
echo      "QP10/QP11 $&procOff    $&Tarea     $&pairOff   $&vingain   $&vinoff"

*#7========Measure_VNO6_effect===========
alter      vno5   dc =    0
alter      vno6   dc =    1m
op
let        procOff = unknown.{$thisData}[1]
let        Tarea   = @qp3[area]
let        Ioutref = unknown.{$thisData}[3]
let        gm      = unknown.{$thisData}[4]
let        vingain  = 1000*(out - Ioutref)/gm
let        pairOff = (1/Tarea)^.5*procOff
let        vinoff  = pairOff*vingain
let        unknown.{$thisData}[15] = vingain  
let        unknown.{$thisData}[16] = vinoff
echo      "QP3/QP4   $&procOff    $&Tarea     $&pairOff   $&vingain   $&vinoff"

*#8========Measure_VNO7_effect===========
alter      vno6   dc =    0
alter      vno7   dc =    1m
op
let        procOff = unknown.{$thisData}[1]



let        Tarea   = @qp6[area]
let        Ioutref = unknown.{$thisData}[3]
let        gm      = unknown.{$thisData}[4]
let        vingain  = 1000*(out - Ioutref)/gm
let        pairOff = (1/Tarea)^.5*procOff
let        vinoff  = pairOff*vingain
let        unknown.{$thisData}[17] = vingain  
let        unknown.{$thisData}[18] = vinoff
echo      "QP6/QP7   $&procOff    $&Tarea     $&pairOff   $&vingain   $&vinoff"

*#9========Measure_VNO8_effect===========
alter      vno7   dc =    0
alter      vno8   dc =    1m
op
let        procOff = unknown.{$thisData}[0]
let        Tarea   = @qn7[area]
let        Ioutref = unknown.{$thisData}[3]
let        gm      = unknown.{$thisData}[4]
let        vingain  = 1000*(out - Ioutref)/gm
let        pairOff = (1/Tarea)^.5*procOff
let        vinoff  = pairOff*vingain
let        unknown.{$thisData}[19] = vingain  
let        unknown.{$thisData}[20] = vinoff
echo      "QN7/QN8   $&procOff    $&Tarea     $&pairOff   $&vingain   $&vinoff"

*#10========Add_Up_total_offset===========
let        off1   = unknown.{$thisData}[6]
let        off2   = unknown.{$thisData}[8]
let        off3   = unknown.{$thisData}[10]
let        off4   = unknown.{$thisData}[12]
let        off5   = unknown.{$thisData}[14]
let        off6   = unknown.{$thisData}[16]
let        off7   = unknown.{$thisData}[18]
let        off8   = unknown.{$thisData}[20]
let        Vin_total =  (off1^2+off2^2+off3^2+off4^2+off5^2+off6^2+off7^2+off8^2 )^.5 

echo       " "
echo       "Total referred to input offset   $&Vin_total"
echo       " "

*#8========Measure_Maximum_Iout===========
alter      vno8   dc =    0
alter      vin    dc =    -430m
op
let        unknown.{$thisData}[22] = out
*echo       "Clipping +/- output current          $&out A"

alter      vin   dc =    0
op

*==========Calculating_Noise_Each_Transitor===================
* Find Area and Emitter current for each transistor (Qn)
* Noise Resistance = (kt/q)/(2*Ie) + Rb/area
* Input_noise_Qn = 4.07nV/rtHz*sqrt(Noise Resistance/1000)
* output_noise_Qn = output_noiseQn*Gain_Qn

*#7========Calc_VNO1_effect_On_Noise===========
let        icc     = @qp1[ic]
let        Tarea   = @qp1[area]
let        rbb     = @PNP1[rb]/Tarea
let        NoiseR  = rbb + 13m/icc
let        NoiseP  = 4.07n*(2*NoiseR/1000)^.5
let        Gain2in = unknown.{$thisData}[5]
let        VinNois = abs(NoiseP*Gain2in)
let        gm      = unknown.{$thisData}[4]
let        IoutNo  = VinNois*gm
let        unknown.{$thisData}[23] = VinNois
echo      "Pair     Ic           NoiseRes  Pair_Noise   Gain2Vin  VnoiseIn_rtHz  OutNoise_rtHz"
echo      "QP1/QP2  $&icc  $&NoiseR   $&NoiseP  $&Gain2in   $&VinNois    $&IoutNo"

*#8========Calc_VNO2_effect_On_Noise===========
let        icc     = @qn1[ic]
let        Tarea   = @qn1[area]
let        rbb     = @NPN1[rb]/Tarea
let        NoiseR  = rbb + 13m/icc
let        NoiseP  = 4.07e-9*(2*NoiseR/1000)^.5
let        Gain2in = unknown.{$thisData}[7]
let        VinNois = abs(NoiseP*Gain2in)
let        gm      = unknown.{$thisData}[4]
let        IoutNo  = VinNois*gm
let        unknown.{$thisData}[24] = VinNois
echo      "QN1/QN6  $&icc  $&NoiseR   $&NoiseP   $&Gain2in  $&VinNois    $&IoutNo"



*#9========Calc_VNO3_effect_On_Noise===========
let        icc     = @qn2[ic]
let        Tarea   = @qn2[area]
let        rbb     = @NPN1[rb]/Tarea
let        NoiseR  = rbb + 13m/icc
let        NoiseP  = 4.07e-9*(2*NoiseR/1000)^.5
let        Gain2in = unknown.{$thisData}[9]
let        VinNois = abs(NoiseP*Gain2in)
let        gm      = unknown.{$thisData}[4]
let        IoutNo  = VinNois*gm
let        unknown.{$thisData}[25] = VinNois
echo      "QN2/QN5  $&icc  $&NoiseR   $&NoiseP $&Gain2in  $&VinNois    $&IoutNo"

*#10=======Calc_VNO4_effect_On_Noise===========
let        icc     = @qn3[ic]
let        Tarea   = @qn3[area]
let        rbb     = @NPN1[rb]/Tarea
let        NoiseR  = rbb + 13m/icc
let        NoiseP  = 4.07e-9*(2*NoiseR/1000)^.5
let        Gain2in = unknown.{$thisData}[11]
let        VinNois = abs(NoiseP*Gain2in)
let        gm      = unknown.{$thisData}[4]
let        IoutNo  = VinNois*gm
let        unknown.{$thisData}[26] = VinNois
echo      "QN3/QN4  $&icc  $&NoiseR   $&NoiseP $&Gain2in  $&VinNois    $&IoutNo"

*#10=======Calc_VNO5_effect_On_Noise===========
let        icc     = @qp10[ic]
let        Tarea   = @qp10[area]
let        rbb     = @PNP1[rb]/Tarea
let        NoiseR  = rbb + 13m/icc
let        NoiseP  = 4.07e-9*(2*NoiseR/1000)^.5
let        Gain2in = unknown.{$thisData}[13]
let        VinNois = abs(NoiseP*Gain2in)
let        gm      = unknown.{$thisData}[4]
let        IoutNo  = VinNois*gm
let        unknown.{$thisData}[27] = VinNois
echo      "QP10/QP11 $&icc  $&NoiseR   $&NoiseP $&Gain2in  $&VinNois    $&IoutNo"

*#10=======Calc_VNO6_effect_On_Noise===========
let        icc     = @qp3[ic]
let        Tarea   = @qp3[area]
let        rbb     = @PNP1[rb]/Tarea
let        NoiseR  = rbb + 13m/icc
let        NoiseP  = 4.07e-9*(2*NoiseR/1000)^.5
let        Gain2in = unknown.{$thisData}[15]
let        VinNois = abs(NoiseP*Gain2in)
let        gm      = unknown.{$thisData}[4]
let        IoutNo  = VinNois*gm
let        unknown.{$thisData}[28] = VinNois
echo      "QP3/QP4  $&icc  $&NoiseR   $&NoiseP $&Gain2in  $&VinNois    $&IoutNo"

*#10=======Calc_VNO7_effect_On_Noise===========
let        icc     = @qp6[ic]
let        Tarea   = @qp6[area]
let        rbb     = @PNP1[rb]/Tarea
let        NoiseR  = rbb + 13m/icc
let        NoiseP  = 4.07e-9*(2*NoiseR/1000)^.5
let        Gain2in = unknown.{$thisData}[17]
let        VinNois = abs(NoiseP*Gain2in)
let        gm      = unknown.{$thisData}[4]
let        IoutNo  = VinNois*gm
let        unknown.{$thisData}[29] = VinNois
echo      "QP6/QP7  $&icc  $&NoiseR   $&NoiseP $&Gain2in  $&VinNois    $&IoutNo"

*#10=======Calc_VNO8_effect_On_Noise===========
let        icc     = @qn7[ic]
let        Tarea   = @qn7[area]
let        rbb     = @PNP1[rb]/Tarea
let        NoiseR  = rbb + 13m/icc
let        NoiseP  = 4.07e-9*(2*NoiseR/1000)^.5
let        Gain2in = unknown.{$thisData}[19]
let        VinNois = abs(NoiseP*Gain2in)
let        gm      = unknown.{$thisData}[4]
let        IoutNo  = VinNois*gm
let        unknown.{$thisData}[30] = VinNois
echo      "QN7/QN8  $&icc  $&NoiseR   $&NoiseP $&Gain2in  $&VinNois    $&IoutNo"

*#11=======Add_Up_All_Noise===========

let        off1   = unknown.{$thisData}[23]
let        off2   = unknown.{$thisData}[24]



let        off3   = unknown.{$thisData}[25]
let        off4   = unknown.{$thisData}[26]
let        off5   = unknown.{$thisData}[27]
let        off6   = unknown.{$thisData}[28]
let        off7   = unknown.{$thisData}[29]
let        off8   = unknown.{$thisData}[30]
let        Vin_T  =  (off1^2+off2^2+off3^2+off4^2+off5^2+off6^2+off7^2+off8^2 )^.5 
let        Out_T  = Vin_T*gm 
let        ibb    = @ibias[dc]
let        ibb    = unknown.{$thisData}[22]
let        SigNo  = 20*log(ibb/(Out_T*141)) -3 

let        Rgm    = 1/unknown.{$thisData}[4]
let        Vin_max = ibb*Rgm

echo       " "
echo       "Rgm  (vin/iout)                      $&Rgm Ohms"
echo       "Total referred to input noise_rtHz   $&Vin_T V"
echo       "Vin maxium                           $&Vin_max V"
echo       "Total output current noise_rtHz      $&Out_T A"
echo       "Clipping +/- output current          $&ibb A"
echo       "Clipping 20-20KHz output S/N         $&SigNo dB"

echo       " "

.endc

*======== models Using Near Perfect Transistors ==================
.model    NPN1   NPN(   BF =   200    VAF =  100    rb=400 )
.model    PNP1   PNP(   BF =   30     VAF =  150    rb=900 )
*======== ====== ====== ====== ====== ====== ====== ====== ======

.end

=====================END_OF_SPICE=======================

To Covert PDF to plain text click below

Circuit: AB_Bias_Audio_OffsetNoise

Process         Expectations  
NPN_offset   =  0.0006  
NPN_beta     =  200  
NPN_rbase    =  400  
PNPL_offset  =  0.0011  
PNPL_beta    =  30  
PNPL_rbase   =  900  
Rbase_offset =  0.005  

Ibias        =  1.6E-07  
Isupply      =  2.00566E-06 

Pair      ProcOff   Area  Pair_Off  Gain2Vin   Vin_Offset 
QP1/QP2   0.0011    1     0.0011    -1.00016   -0.00110018
QN1/QN6   0.0006    4.5   0.00028    0.490909   0.00013885
QN2/QN5   0.0006    1     0.0006     0.210912   0.000126547
QN3/QN4   0.0006    15    0.000154  -0.71612   -0.000110941
QP10/QP11 0.0011    1     0.0011     0.872016   0.000959218
QP3/QP4   0.0011    1     0.0011    -1.38178   -0.00151996
QP6/QP7   0.0011    1     0.0011     1.38921    0.00152813
QN7/QN8   0.0006    1     0.0006     1.40827    0.000844963
 
Total referred to input offset   0.00274545
 
Pair      Ic           NoiseRes  Pair_Noise   Gain2Vin  VnoiseIn_rtHz  OutNoise_rtHz
QP1/QP2   5.18398E-07  25977.3   2.93363E-08  -1.00016   2.93412E-08    3.01488E-13
QN1/QN6   1.17124E-07  111083    6.06642E-08   0.490909  2.97806E-08    3.06002E-13
QN2/QN5   2.60279E-08  499865    1.28687E-07   0.210912  2.71417E-08    2.78887E-13
QN3/QN4   3.89535E-07  33399.8   3.32645E-08  -0.71612   2.38214E-08    2.4477E-13
QP10/QP11 1.4582E-07   90051.3   5.46203E-08   0.872016  4.76298E-08    4.89407E-13
QP3/QP4   3.76745E-07  35406.1   3.42491E-08  -1.38178   4.73247E-08    4.86272E-13
QP6/QP7   3.76675E-07  35412.5   3.42521E-08   1.38921   4.75833E-08    4.88929E-13
QN7/QN8   3.85655E-07  34608.9   3.38613E-08   1.40827   4.76858E-08    4.89983E-13
 
Rgm  (vin/iout)                      97321.4 Ohms
Total referred to input noise_rtHz   1.09992E-07 V
Vin maxium                           0.196217 V
Total output current noise_rtHz      1.13019E-12 A



Clipping +/- output current          2.01617E-06 A
Clipping 20-20KHz output S/N         79.0431 dB

=================AB_Bias_Input_Comparision=========================

Since the signal to noise ratio is supply current dependent,
the comparison needs to be made at a common 2uA supply current
level. This simulation allows the introduction of expected 
process parameters such as base resistance, beta, and offset. 
Spice's "alter" feature is used to find individually the 
contributions of each and every matched pair to the over all 
performance.  

The comparing the AB_Bias Input OTA to a LM3080 Input OTA

                         AB_BIAS         LM3080

1)  measure supply current        = 2.045uA              2.0uA
2)  Measure Max output current    = +/- 2.016uA_pk       +/- .97uA_pk
3)  Measure Max Vin extrapolate   = +/- 196mV_pk         +/- 51mV_pk
4)  Measure Offset_V              = +/- 2.7mV_sd         +/- 1.6mV_sd
5)  Measure Offset_Iout_%Imax     = +/- 1.36_%           +/- 3.2_%
6)  Measure input noise_rtHz      = +/- 110nV_rms        +/- 56nV_rms 
7)  Measure 1% THD output current = +/- 1.50uA_pk        +/- .334uA_pk
8)  Measure 1% THD input V        = +/- 146mV_pk         +/- 18mV_pk
9)  Meaured Rgm = Vin/Iout        = 97KOhms              51KOhms
10) Total Output Shot Noise       = 159pA_rms            151pA_rms
11) Signal to Noise Clipping      = 79.0dB               73dB
12) Signal to Noise 1% THD        = 76.4dB               63.5dB

The maximum output current of the AB_bias circuit
is twice that of the normal input stage. Since there are
a lot more transistors in the AB_Bias input stage, both
the referred-to input noise and offset will have to increase.
But since the AB_Bias stage can handle much higher signal
levels, there is about a factor of two improvement in both
offset and noise when operating at the 100% signal levels.
With distortion an issue, the AB_Bias input can run at
about 75% at a 1% distortion. The Dual differential input
will be at the 62% level. A non pre-distorted LM3080 is at
a 35% level. In theory a perfect pre-distortion stage can
take a LM3080 up to a 100% level. But even when the 
supply current of a perfect pre-distortion stage is high enough
so that it will not effect the signal to noise ratio, 
the signal to noise ratio will still be 3dB lower than for the 
AB_Bias Input stage. 




